Survival Models in SAS

Part 5: PROC LIFEREG – Part 3

February 13, 2008

Charlie Hallahan

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG

These talks are based on the book "**Survival Analysis Using the SAS System: A Practical Guide**" (1995) by Paul Allison.

The book is part of the SAS Books-by-Users series and can be found at http://www.sas.com/apps/pubscat/bookdetails.jsp?catid=1&pc=55233



Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG

This series of talks will cover

Chapter 1: Introduction

Chapter 2: Basic Concepts of Survival Analysis

Chapter 3: Estimating and Comparing Survival Curves with PROC LIFETEST

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG

Chapter 5: Estimating Cox Regression Models with PROC PHREG

Chapter 6: Competing Risks

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG

Topics in Chapter 4:

Introduction
The Accelerated Failure Time Model
Alternative Distributions
Categorical Variables and the CLASS Statement
Maximum Likelihood Estimation
Hypothesis tests
Goodness-of-Fit Tests with the Likelihood-Ratio Statistic
Graphical Methods of Evaluating Model Fit
Left Censoring and Interval Censoring
Generating Predictions and Hazard Functions
The Piecewise Exponential Model

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Left Censoring and Interval Censoring

A feature of **LIFEREG,** not shared by **PHREG**, is its ability to handle both left censoring and interval censoring.

Left censoring means the event of interest occurred sometime earlier than the observed time *t*.

Interval censoring means that the event of interest occurred sometime in an interval [a, b], but it is not known exactly when in that interval.

Left censoring is a special case of interval censoring where a = 0.

Right censoring is a special case of interval censoring where b = ∞.

Interval censoring is incorporated into the likelihood function as $S_i(a) - S_i(b)$, where $S_i(.)$ is the survivor function for observation *i*.

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Left Censoring and Interval Censoring

A different syntax for the **MODEL** statement is required when either left or interval censoring is present.

MODEL(LOWER,UPPER) = *list of covariates*

The form of censoring for an observation depend on the values for UPPER and LOWER.

Uncensored:	LOWER and UPPER are both present and equal
Interval Censored:	LOWER and UPPER are both present and different
Right Censored:	LOWER is present and UPPER is missing
Left Censored:	LOWER is missing and UPPER is present

Any observation where LOWER is 0 or negative or both UPPER and LOWER are missing are excluded.

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Left Censoring and Interval Censoring

For example:

Observation	Lower	Upper	Status .
1	3.9	3.9	Uncensored
2	7.2	.	Right Censored
3	4.1	5.6	Interval Censored
4	.	2.0	Left Censored
5	0	5.8	Excluded
6	3.2	1.9	Excluded

Using the recidivism dataset, we'll artificially create censored data of various kinds.

Recall that the current dataset only contains right-censored observations where no arrest has occurred by week = 52, the end of the study.

The first 30 arrests are recoded to have **WEEK** equal to missing.

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Left Censoring and Interval Censoring

```
data recid_temp;
   retain count 0;
   set survival.recid;

   * set week = . for 1st 30 arrests;
   if count < 30 and arrest = 1 then
          do;
                    week = .;
                    count = count + 1;
          end;

   * artificially create some censored data;
   if _n_ <= 30 then week = .;

       /* uncensored cases: */
   if arrest=1 and week ne . then do;
      upper=week;
      lower=week;
   end;

      /* left-censored cases: */
   if arrest=1 and week = . then do;
      upper=52;
      lower=.;
   end;

      /* right-censored cases: */
   if arrest=0 then do;
      upper=.;
      lower=52;
   end;
run;
```

← creating some left-censored obs.

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Left Censoring and Interval Censoring

```
title "Example with right- and left-censoring";
proc lifereg data = recid_temp;
   model (lower,upper)=fin age race wexp mar paro prio
         / dist=weibull;
run;
```

Model Information

Data Set	WORK.RECID_TEMP
Dependent Variable	Log(lower)
Dependent Variable	Log(upper)
Number of Observations	**432**
Noncensored Values	**84**
Right Censored Values	**318**
Left Censored Values	**30**
Interval Censored Values	**0**
Name of Distribution	Weibull
Log Likelihood	-294.0611169

Number of Observations Read	432
Number of Observations Used	432

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Left Censoring and Interval Censoring

Analysis of Parameter Estimates

Parameter	DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq
Intercept	1	3.9567	0.4078	3.1574	4.7560	94.14	<.0001
fin	1	0.2490	0.1348	-0.0152	0.5133	3.41	0.0647
age	1	0.0413	0.0158	0.0102	0.0723	6.79	0.0092
race	1	-0.2191	0.2145	-0.6395	0.2013	1.04	0.3070
wexp	1	0.0777	0.1484	-0.2132	0.3686	0.27	0.6006
mar	1	0.3001	0.2658	-0.2208	0.8211	1.27	0.2588
paro	1	0.0682	0.1356	-0.1975	0.3339	0.25	0.6151
prio	1	-0.0623	0.0207	-0.1030	-0.0217	9.03	0.0027
Scale	1	0.6911	0.0717	0.5639	0.8470		
Weibull Shape	1	1.4469	0.1501	1.1806	1.7732		

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Left Censoring and Interval Censoring

For comparison, here is the **original model** (without left-censoring):

Analysis of Parameter Estimates

Parameter	DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq
Intercept	1	3.9901	0.4191	3.1687	4.8115	90.65	<.0001
fin	1	0.2722	0.1380	0.0018	0.5426	3.89	0.0485
age	1	0.0407	0.0160	0.0093	0.0721	6.47	0.0110
race	1	-0.2248	0.2202	-0.6563	0.2067	1.04	0.3072
wexp	1	0.1066	0.1515	-0.1905	0.4036	0.49	0.4820
mar	1	0.3113	0.2733	-0.2244	0.8469	1.30	0.2547
paro	1	0.0588	0.1396	-0.2149	0.3325	0.18	0.6735
prio	1	-0.0658	0.0209	-0.1069	-0.0248	9.88	0.0017
Scale	1	0.7124	0.0634	0.5983	0.8482		
Weibull Shape	1	1.4037	0.1250	1.1789	1.6713		

Note that the parameter estimates are close, but that the chi-square values are nearly all smaller for the model with the 30 left-censored observations, reflecting the decreased information in the data.

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Left Censoring and Interval Censoring

The data for the recidivism example is recorded weekly. This could be considered as daily data which has been interval-censored.

The following code creates a new dataset where the arrest data is treated as daily data which has been interval-censored as weekly data.

```
data recid_temp;
  set survival.recid;

     /* interval censored cases: */
   if arrest=1 then do;
      upper=week;
      lower=week-.9999;
   end;

     /* right-censored cases: */
   if arrest=0 then do;
      upper=.;
      lower=52;
   end;
```

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Left Censoring and Interval Censoring

```
title "Example with interval-censoring";
proc lifereg data = recid_temp;
   model (lower, upper)=fin age race wexp mar paro prio
         / dist=weibull;
run;
```

Model Information

Data Set	WORK.RECID_TEMP
Dependent Variable	Log(lower)
Dependent Variable	Log(upper)
Number of Observations	**432**
Noncensored Values	**0**
Right Censored Values	**318**
Left Censored Values	**0**
Interval Censored Values	**114**
Name of Distribution	Weibull
Log Likelihood	-680.995873

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Left Censoring and Interval Censoring

Analysis of Parameter Estimates

Parameter	DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq
Intercept	1	3.9906	0.4374	3.1333	4.8479	83.23	<.0001
fin	1	0.2837	0.1440	0.0015	0.5659	3.88	0.0488
age	1	0.0425	0.0167	0.0098	0.0753	6.48	0.0109
race	1	-0.2343	0.2298	-0.6846	0.2161	1.04	0.3079
wexp	1	0.1106	0.1582	-0.1994	0.4206	0.49	0.4845
mar	1	0.3246	0.2853	-0.2345	0.8837	1.29	0.2552
paro	1	0.0618	0.1457	-0.2238	0.3474	0.18	0.6714
prio	1	-0.0685	0.0218	-0.1113	-0.0257	9.83	0.0017
Scale	1	0.7435	0.0665	0.6239	0.8861		
Weibull Shape	1	1.3449	0.1204	1.1285	1.6028		

Again the parameter estimates are close to those with just the right-censoring, but that the chi-square values are nearly all smaller for this model with the 114 interval-censored observations, reflecting the decreased information in the data.

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

When predicting a single survival time for a particular observation, the **median survival time** is usually considered the most useful.

The **OUTPUT** statement produces such a prediction.

```
title "Outputting the median survival time for each observation";
proc lifereg data=survival.recid;
   model week*arrest(0)=fin age race wexp mar paro prio
         / dist=weibull;
   output out=a p=median std=s;
run;

proc print data=a(obs=20);
   var week arrest _prob_ median s;
run;
```

← the **P=** keyword produces the median by default. Other percentiles can be requested with the **QUANTILE** keyword. **STD=** requests the standard errors of the median predictions.

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

Obs	week	arrest	_PROB_	median	s
1	20	1	0.5	86.910	14.5322
2	17	1	0.5	43.353	6.6456
3	25	1	0.5	45.257	14.6537
4	52	0	0.5	167.947	50.2931
5	52	0	0.5	87.403	21.7306
6	52	0	0.5	86.166	13.9891
7	23	1	0.5	148.225	42.4156
8	52	0	0.5	93.083	15.5320
9	52	0	0.5	54.873	7.6941
10	52	0	0.5	83.518	15.4830
11	52	0	0.5	109.544	18.9859
12	52	0	0.5	165.290	50.3119
13	37	1	0.5	56.411	10.1319
14	52	0	0.5	146.285	38.7677
15	25	1	0.5	65.358	8.3600
16	46	1	0.5	110.592	18.1147
17	28	1	0.5	45.471	6.8145
18	52	0	0.5	65.816	10.3064
19	52	0	0.5	44.305	9.4288
20	52	0	0.5	122.208	30.5519

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

To get predicted values for observations not in the original dataset, simply add these observations to the original dataset with a missing values for the event time.

To restrict predictions to just a subset of observations, create a dummy variable, say **`USE`**, that equals 1 for a prediction observation and 0 otherwise.

Then use the option **`CONTROL=USE`** in the **`OUTPUT`** statement.

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

To request specific predicted survival times, the author provides a macro called **`PREDICT`**.

The calculations are straightforward once the linear predictor for an observation and parameter estimates are known.

These are available in **`PROC LIFEREG`** with the **`OUTEST`** option on the **`PROC`** statement and the **`XBETA=`** option on the **`OUTPUT`** statement.

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

```
*--------------;
* PREDICT Macro;
*--------------;

%macro predict(outest=, out=_last_,xbeta=,time=);

/********************************************************************
MACRO PREDICT produces predicted survival probabilities for specified
survival times, based on models fitted by PROC LIFEREG. When fitting the
model with PROC LIFEREG, you must request the OUTEST data set in the
PROC statement.  You must also request an OUTPUT data set with the
XBETA= keyword.

PREDICT has four parameters:

OUTEST    is the name of the data set produced with the OUTEST option.

OUT       is the name of the data set produced by the OUTPUT statement.
          The default is the last created data set.

XBETA     is the name of the variable specified with the XBETA= keyword.

TIME      is the specified survival time that is to be evaluated.
```

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

```
Example of usage:

To get 5-year survival probabilities for every individual in the sample
(assuming that actual survival times are measured in years);

%predict(outest=a, out=b, xbeta=lp, time=5).

Author:  Paul D. Allison, Univ. of Pennsylvania
         allison@ssc.upenn.edu

               (edited 10/25/2007)
**********************************************************************/
```

(I had to make a few edits to be compatible with the current version of SAS)

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

```
data _pred_;
         _p_=1;
         * use if _dist_ = Gamma;
         *set &outest (keep=_dist_ _scale_ _shape1_ ) point=_p_;
         set &outest (keep=_dist_ _scale_ ) point=_p_;
         file print;
         set &out;
         lp=&xbeta;
         t=&time;
         gamma=1/_scale_;
         alpha=exp(-lp*gamma);
         prob=0;
         if _dist_='Exponential' or _dist_='Weibull' then prob=exp(-alpha*t**gamma);
         if _dist_='Lognormal' then prob=1-probnorm((log(t)-lp)/_scale_);
         if _dist_='LLogistic' then prob=1/(1+alpha*t**gamma);
         if _dist_='Gamma' then
                  do;
                              d=_shape1_;
                              k=1/(d*d);
                              u=(t*exp(-lp))**gamma;
                              prob=1-probgam(k*u**d,k);
                              if d lt 0 then prob=1-prob;
                   end;
         drop lp gamma alpha _dist_ _scale_ d k u;
run;
%mend predict;
```

(The variable **_shape1_** only exists for the gamma distribution)

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

```
%include "d:\Charlie\SIGSTAT\Survival Models\predict.sas";

title "Predicting 30-week survival probabilities";
proc lifereg data=survival.recid outest=a;
   model week*arrest(0)=fin age race wexp mar paro prio
         / dist=weibull;
   output out=b p=median xbeta=lp;
run;

%predict(outest=a, out=b, xbeta=lp, time=30)

title2 "Summary statistics for 30-week predicted  survival probabilities";
proc means data=_pred_;
          var prob;
run;

title2 "1st 20 observations with 30-week predicted  survival probabilities";
proc print data=_pred_(obs=20);
          var week arrest fin age race wexp mar paro prio t prob;
run;
```

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

Summary statistics for 30-week predicted survival probabilities

The MEANS Procedure

Analysis Variable : **prob**

N	Mean	Std Dev	Minimum	Maximum
432	0.8639191	0.0749853	0.3436233	0.9831820

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

1st 20 observations with 30-week predicted survival probabilities

Obs	week	arrest	fin	age	race	wexp	mar	paro	prio	t	**prob**
1	20	1	0	27	1	0	0	1	3	30	0.85579
2	17	1	0	18	1	0	0	1	8	30	0.66139
3	25	1	0	19	0	1	0	1	13	30	0.67760
4	52	0	1	23	1	1	1	1	1	30	0.94010
5	52	0	0	19	0	1	0	1	3	30	0.85684
6	52	0	0	24	1	1	0	0	2	30	0.85417
7	23	1	0	25	1	1	1	1	0	30	0.92903
8	52	0	1	21	1	1	0	1	4	30	0.86811
9	52	0	0	22	1	0	0	0	6	30	0.74306
10	52	0	0	20	1	1	0	0	0	30	0.84816
11	52	0	1	26	1	0	0	1	3	30	0.89356
12	52	0	0	40	1	1	0	0	2	30	0.93878
13	37	1	0	17	1	1	0	1	5	30	0.75151
14	52	0	0	37	1	1	0	0	2	30	0.92776
15	25	1	0	20	1	0	0	1	3	30	0.79267
16	46	1	1	22	1	1	0	1	2	30	0.89490
17	28	1	0	19	1	0	0	0	7	30	0.67934
18	52	0	0	20	1	0	0	0	2	30	0.79448
19	52	0	0	25	1	0	0	1	12	30	0.66965
20	52	0	0	24	0	1	0	1	1	30	0.90800

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

The author provides another macro, **LIFEHAZ**, to plot the estimated hazard function appropriate for the selected model distribution.

```
*---------------;
* LIFEHAZ Macro ;
*---------------;

%macro lifehaz(outest=,out=,obsno=0,xbeta=lp);

/*********************************************************************
Macro LIFEHAZ plots the hazard function for a model fitted by
PROC LIFEREG. In the LIFEREG procedure, you must specify OUTEST=name1
in the PROC statement.  You must also use the OUTPUT statement with
OUT=name2 and XBETA=name3. By default, the hazard is plotted for the
mean value of XBETA (the linear predictor).  If you want a plot for a
specific observation, you must specify the observation number
(OBSNO) when you invoke the macro.  The macro is invoked as follows:

   %lifehaz(outest=name1,out=name2,xbeta=name3,obsno=1);

Author: Paul D. Allison, U. of Pennsylvania, allison@ssc.upenn.edu.

                              (edited:  10/25/2007)

*********************************************************************/
```

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

```
data;
  set &outest;
  call symput('time',_NAME_);
run;
proc means data=&out noprint;
  var &time &xbeta;
  output out=_c_ min(&time)=min max(&time)=max mean(&xbeta)=mean;
run;
data;
  set &outest;
  call symput('model',_dist_);
  s=_scale_;
  *d=_shape1_;
  _y_=&obsno;
  set _c_ (keep=min max mean);
  if _y_=0 then m=mean;
  else do;
    set &out (keep=&xbeta) point=_y_;
    m=&xbeta;
  end;
  inc=(max-min)/300;
  g=1/s;
  alph=exp(-m*g);
```

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

```
if _dist_='LNORMAL' then do;
  do t=min to max by inc;
  z=(log(t)-m)/s;
  f=exp(-z*z/2)/(t*s*sqrt(2*3.14159));
  Surv=1-probnorm(z);
  h=f/Surv;
  output;
  end;
end;
else if _dist_='Gamma' then do;
  *d=_shape1_;
  k=1/(d*d);
  do t=min to max by inc;
  u=(t*exp(-m))**(1/s);
  f=abs(d)*(k*u**d)**k*exp(-k*u**d)/(s*gamma(k)*t);
  Surv=1-probgam(k*u**d,k);
  if d lt 0 then Surv=1-Surv;
  h=f/Surv;
  output;
  end;
end;
else if _dist_='Weibull' or _dist_='Exponential' then do;
  do t=min to max by inc;
  h=g*alph*t**(g-1);
  output;
  end;
end;
```

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions

```
          else if _dist_='LLogistic' then do;
            do t=min to max by inc;
            h=g*alph*t**(g-1)/(1+alph*t**g);
            output;
            end;
          end;
          else put 'ERROR:DISTRIBUTION NOT FITTED BY LIFEREG';
          run;
          proc gplot;
            plot h*t / haxis=axis2 vaxis=axis1 vzero;
            symbol1 i=join v=none c=black;
            axis1 label=(f=titalic angle=90 'Hazard');
            axis2 label=(f=titalic justify=c 'time' f=titalic justify=c "&model");
          run; quit;
%mend lifehaz;
```

```
%include "d:\Charlie\SIGSTAT\Survival Models\lifehaz.sas";
title "Graph of Estimated Hazard Function for the Recidivism Data";
title2 "Estimated Parametric Model is Weibull";
%lifehaz(outest=a, out=b, xbeta=lp)
```

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: Generating Predictions and Hazard Functions



Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: The Piecewise Exponential Model

All the parametric models considered so far assume a smooth, relatively simple function of time.

The **Cox** model (estimated with **PROC PHREG**) makes no such assumptions. The cost for this flexibility is not being able to test the shape of the hazard function.

A compromise to introduce some flexibility into a parametric survival model is the **piecewise exponential model**.

To some extent this model incorporates some **time-varying covariates**, namely the intercept is allowed to vary over time.

Some **preliminary data manipulation** is necessary with the existing data.

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: The Piecewise Exponential Model

Basic idea:

- Divide the time scale into intervals
- Assume the hazard is constant within each interval, but can vary across intervals.

Let $a_0, a_1, \ldots, a_J$ be the endpoints for the J intervals with $a_0 = 0$ and $a_J = \infty$.

The **hazard** in the interval $[a_{j\text{-}1}, a_j)$ is assumed to be: $h_j(t) = \lambda_j e^{\boldsymbol{\beta}\mathbf{x}}$

or $\log\left(h_j(t)\right) = \alpha_j + \boldsymbol{\beta}\mathbf{x}$ where $\alpha_j = \log\left(\lambda_j\right)$.

The choice of intervals is arbitrary (implying some ad hoc ness for the model).

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: The Piecewise Exponential Model

As an example, consider the **recidivism** data where a different intercept is allowed for the log hazard for each quarter of the year.

The 52-week year is divided into 4 13-week quarters and a separate observation is created for each subject for any quarter they are still at risk, i.e., haven't been arrested yet by the beginning of that quarter.

So four records are created for any subject not arrested by the beginning of the fourth quarter, three records for those not arrested by the beginning of the third quarter, etc..

The recidivism data is easier than most survival datasets since all censoring takes place at the end of the dataset in the fourth quarter. Otherwise the code below would be more complicated in that it would have to check within each quarter whether or not an observation is censored or not in order to correctly define the censoring variable.

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: The Piecewise Exponential Model

```
data quarter;
   set survival.recid;
   quarter=ceil(week/13);
   do j=1 to quarter;
      time=13;
      event=0;
        if j=quarter and arrest=1 then do;
                event=1;
                time=week-13*(quarter-1);
        end;
        output;
   end;
run;
```

Note that the variable `time` now represents the number of weeks within each quarter that the subject is either at risk or has been arrested (i.e., the event has occurred).

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: The Piecewise Exponential Model

```
title "Estimation of a piecewise exponential model";
proc lifereg data=quarter;
   class j;
   model time*event(0)=fin age race wexp mar paro prio j
         / dist=exponential covb;
run;
```

Type III Analysis of Effects

Effect	DF	Wald Chi-Square	Pr > ChiSq
fin	1	3.8895	0.0486
age	1	6.7348	0.0095
race	1	1.0302	0.3101
wexp	1	0.4924	0.4828
mar	1	1.2869	0.2566
paro	1	0.1826	0.6692
prio	1	10.0699	0.0015
j	**3**	**8.6954**	**0.0336**

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: The Piecewise Exponential Model

Analysis of Parameter Estimates

Parameter		DF	Estimate	Standard Error	95% Confidence Limits		Chi-Square	Pr > ChiSq
Intercept		1	3.7226	0.6082	2.5307	4.9146	37.47	<.0001
fin		1	0.3774	0.1913	0.0023	0.7524	3.89	0.0486
age		1	0.0570	0.0220	0.0140	0.1000	6.73	0.0095
race		1	-0.3126	0.3080	-0.9162	0.2910	1.03	0.3101
wexp		1	0.1489	0.2122	-0.2670	0.5648	0.49	0.4828
mar		1	0.4331	0.3818	-0.3152	1.1814	1.29	0.2566
paro		1	0.0836	0.1957	-0.3000	0.4672	0.18	0.6692
prio		1	-0.0909	0.0286	-0.1470	-0.0347	10.07	0.0015
j	**1**	**1**	**0.8202**	**0.2841**	**0.2633**	**1.3771**	**8.33**	**0.0039**
j	**2**	**1**	**0.1883**	**0.2446**	**-0.2911**	**0.6677**	**0.59**	**0.4414**
j	**3**	**1**	**0.3134**	**0.2596**	**-0.1953**	**0.8221**	**1.46**	**0.2273**
j	4	0	0.0000	.	.	.	.	.
Scale		0	1.0000	0.0000	1.0000	1.0000		
Weibull Shape		0	1.0000	0.0000	1.0000	1.0000		

So there is a significant effect of J (quarter), especially in the first quarter.

Chapter 4: Estimating Parametric Regression Models with PROC LIFEREG: The Piecewise Exponential Model

Recall the **LIFEREG** estimates the model $\log(T_i) = x_i\beta + \sigma\varepsilon_i$ so that a positive parameter estimate implies a longer survival time for subject *i*.

A longer survival time corresponds to a lower hazard and the sign of the coefficient must be reversed to interpret the impact of that variable on the hazard function at T_i.

Thus, an significant estimate of 0.8202 for the first quarter effect implies that, relative to the fourth quarter, there is less hazard for arrest in the first quarter.

The second and third quarter effects are also positive, but insignificant.